
Ⅰ Fortran 90의 개요
1950년대 후반 IBM에서 개발된 Fortran은 가장 오래된 고급언어로, 처음에는 IBM Mathematical FORmula TRANslation System이란 긴 이름을 가지고 있었으며, 이후 줄여서FORmular TRANslation, 즉 Fortran이라 불리게 되었다. Fortran 표준은 1970년대 후반에 ANSI에 의해 새로운 표준으로 업데이트 되었고, 1980년도에 ISO에 의해 국제표준으로 채택되었는데 이것이 지금까지 널리 사용되어지는 Fortran77 이다(초안이 1977년에 완성 되었기 때문에 77이란 이름이 붙었다). 

Fortran90은 Fortran77을 획기적으로 발전시킨 것이지만 Fortran77의 모든 것을 포함하고 있다. 그래서 Fortran77 표준을 따르는 어떤 프로그램도 Fortran90 프로그램으로 사용될 수 있다. Fortran90은 Fortran77의 모든 기능에 프로그램 작성을 보다 편리하게 하는 다수의 새로운 특징들을 포함하고 있으며, 특히 객체 지향의 프로그래밍이 가능하고 현재 개발된 프로그래밍 언어 중에서 배열을 다루기에 가장 편리한 고수준의 문법을 갖추고 있다.

본 기술서에서는 앞으로 3회에 걸쳐 Fortran77에서 새롭게 추가된 것들을 중심으로 계산과학 분야에 주로 사용되는 Fortran90 프로그래밍 언어의 특징과 기능들에 초점을 맞춰 소개할 것이다. 

1. Fortran77의 단점

 Fortran77은 다음과 같은 단점을 가지고 있다. 

· Fortran77은 “천공카드” 또는 “고정형식”의 소스코드 포맷을 가진다.

· Fortran77 프로그램의 각 줄은 72열까지만 허용된다. 

· 각 줄의 첫 5열은 줄 번호를 적는 곳이다. 

· 각 줄의 제 6열은 줄 연결을 나타내는 것만으로 사용된다.

· 주석은 반드시 첫 1열에서 특정 문자로 시작되어야 하고 프로그램 코드가 쓰여진 줄에 삽입되어 같이 쓰여질 수 없다. 

· 변수 이름이 6글자로 제한된다.

· 본질적으로 병렬 연산을 표현할 수 없다.

· 동적 저장을 지원하지 않는다. 

· 수치적 이식성 부족 : Fortran77 코드를 다른 시스템으로 포팅하는 경우 각 시스템간의 정밀도 표현이 달라 문제가 발생한다.

· 새로운 자료구조를 사용자가 정의해 사용할 수 없다.

· 재귀적 호출(자기 자신을 호출해 사용)을 지원하지 않는다.  

· COMMON 블록, EQUIVALENCE 문 등의 불안정성

2. 새로운 기능 소개

그림 1. Fortran90 (출처:http://www.comphys.uni-duisburg.de/Fortran90/pl/pl.html)

그림 1은 Fortran90을 구성하는 주요 내용들을 나타내고 있다. Fortran90은 Fortran77에 새로운 기능들이 추가된 것이다. Fortran90에서 새롭게 추가된 특성을 여기서 간략히 소개한다. 소개되는 내용에 대한 보다 자세한 설명과 사용법, 예제 등은 3, 4, 5장에서 다루게 될 것이다.  

1) 개선된 문법 

· 자유형식 - 어떤 위치에서 문장이 시작되어도 상관없다. 

· 한 줄에 132열까지 사용 가능

· 한 줄에 한 문장 이상 가능 – 문장 분리 기호 ‘;’
· 문장 내에 주석을 달 수 있음 – 주석 시작 기호 ‘!’
· 줄 바꿈 표시 기호 ‘&’
· 변수 이름 31자까지 사용 가능 
 (반드시 문자로 시작되어야 한다.)

· 관계 연산자 .LT., .LE., .EQ., .NE., .GE., .GT.를 각각 <, <=, ==, /=, =>, >로 대체 가능
	PRINT*, "This line is continued  & 

            on the next line"; END ! of program



※ Fortran90에서 사용 가능한 문자 집합

· 문자-숫자 조합 : a-z, A-Z, 0-9, _ (underscore)

· 기호(Fortran90에서 새로 추가된 기호는 굵게 표시)  

	기 호
	기   술
	기 호
	기   술

	
	space
	=
	equal

	+
	plus
	-
	minus

	*
	asterisk
	/
	slash

	(
	left parenthesis
	)
	right parenthesis

	,
	comma
	.
	period

	‘
	single quote
	“
	double quote

	:
	colon
	;
	semi colon

	!
	shriek
	&
	ampersand

	%
	percent
	<
	less than

	>
	greater than
	$
	dollar

	?
	question mark
	
	


2) 새로운 방식의 Type Declarations과 Attributing 

	Fortran77
	Fortran90

	FUNCTION encode(pixels)

INTEGER n

PARAMETER(n=1000) 

INTEGER pixels(n, n), encode(n, n) 

REAL x(n), y(n) 

INTEGER*8 call
call = 0 

...

END

	FUNCTION encode(pixels)

IMPLICIT NONE

INTEGER, PARAMETER :: n=1000 

INTEGER, DIMENSION(n, n) :: pixels, encode 

REAL, DIMENSION(n) :: x, y 

INTEGER(8) :: call=0 

...

END


3) 새로운 Control Structures

· 정수 또는 문자 표현식에 대한 SELECT-CASE문 

· IF문 같은 순차적 선택이 아닌 병렬적 선택이 가능하도록 한다. 

	SELECT CASE (expression)
CASE(value-list)
…

CASE(value-list)
…

END SELECT


	PROGRAM select

INTEGER :: n, k

PRINT *, 'Enter the value n = '

READ *, n

SELECT CASE(n)
CASE (:0)

     k = -n
CASE (10:)

     k = n+10

CASE DEFAULT
     k = n
END SELECT
PRINT *, k

END




· 인덱스 없는 새로운 DO 구문

· DO - END DO

· DO WHILE – END DO

· 인덱스에 의한 계산을 배열 연산으로 대체함으로써 데이터 병렬성을 가지는 계산에서 성능 이득을 얻을 수 있다.  

	DO

…
IF (logical-expr) EXIT

…

END DO


	DO WHILE (logical-expr)

…
END DO


4) Numeric Processing

여러 유용한 모델 값을 리턴하는 고유 함수들과 kind 값을 리턴하는 고유 함수들, 그리고 기타 일반적인 연산들에 대해 2장에서 다룰 것이다. 

5) Array Processing

배열 연산은 Fortran90의 가장 중요한 부분이며 이는 4장에서 자세하게 다룰 것이다.

6) Pointers

Fortran90에서 포인터는 동적 자료 구조와 동적 배열이라는 두 가지 중요한 성능을 제공한다. Fortran90의 포인터는 포인터 target으로 명시적으로 선언된 데이터 object, 동적으로 생성된 object 또는 다른 포인터를 가리킨다(point to). 관련 내용은 5장에서 자세히 다룰 것이다. 

	REAL, TARGET  :: B(100,100)         ! 배열 B는 target 속성을 가진다.
REAL, POINTER :: U(:,:),V(:),W(:,:)    ! 3개의 포인터 배열 선언
  ...

U => B(I:I+2,J:J+2)                 ! U는 B의 3x3 부분을 point
ALLOCATE ( W(M,N) )                ! 크기가 MxN인 W를 동적할당

V => B(:,J)                         ! V는 B의 J번째 열을 point 
V => W(I-1,1:N:2)                  ! V는 W의 I-1번째 열의 일부를 point하도록 바뀜 



7) 사용자 정의 타입과 연산자

새로운 사용자 정의 타입은 기존의 데이터 타입을 이용해 정의되며, TYPE – END TYPE 구문을 사용한다. 예를 들어, 하나의 유리수를 분자와 분모로 표현하는 새로운 타입을 정의하고 이러한 타입을 가지는 변수들을 다음과 같이 선언할 수 있다. 

	TYPE RATIONAL                     ! This defines the type RATIONAL.

INTEGER :: NUMERATOR
INTEGER :: DENOMINATOR

END TYPE RATIONAL

  ...

TYPE (RATIONAL) :: X, Y(100,100)  ! X and Y are variables of type RATIONAL.
…
X%NUMERATOR = 2; X%DENOMINATOR = 3




위와 같이 새롭게 정의된 타입을 가지는 변수들의 연산을 위해서 적절한 연산들을 정의할 필요가 있다. 사용자 정의 연산자는 사용자가 만든 함수와 이에 대한 연산자 인터페이스의 정의로 표현된다. 다음 예는 데이터 타입 RATIONAL로 정의된 변수 사이의 덧셈을 위해 확장된 “+” 연산자의 인터페이스를 정의한 것이다. 

	INTERFACE OPERATOR(+)

FUNCTION RAT_ADD(X, Y)

  TYPE (RATIONAL) :: RAT_ADD

TYPE (RATIONAL), INTENT(IN) :: X, Y

END FUNCTION RAT_ADD

END INTERFACE




사용자 정의 데이터 타입과 연산자에 대한 보다 자세한 내용은 5장에서 다룰 것이다.

8) 자료구조

배열은 과학계산 분야에서 가장 중요하고 가장 많이 사용되는 자료형이지만 동적연결구조(Dynamically Linked Structures)와 같은 다른 형식의 자료구조 또한 필요하다. 다음 예는 동적연결구조를 가지는 데이터 타입을 정의한 것이다. Fortran90에서, 자료구조는 사용자 정의 타입을 가지는 객체(혹은 변수)로 선언되어 사용될 수 있다. 

	TYPE LIST

REAL                :: DATA

TYPE (LIST), POINTER :: PREVIOUS, NEXT   !Recursive Components 

END TYPE LIST




9) 프로시저

Fortran90 프로시저는 다음과 같은 새로운 특성을 가진다. 

· 함수의 결과 값이 배열 또는 다른 자료구조가 될 수 있다. 

· 자기 자신을 호출하는 recursive 계산이 가능하다.

· 프로시저 인터페이스를 명시적으로 정의할 수 있다. 

· 서브루틴내에 새로운 서브루틴을 두는 내장 프로시저 사용이 가능하다.

· 인수 사용에 대해 OPTIONAL, INTENT(IN) 등의 속성을 지정할 수 있다.

10) 모듈

모듈은 Fortran90에서 지원하는 프로그램의 새로운 단위로서 주 프로그램(main program) 또는 외부 부 프로그램(external subprogram)과 같은 실행 프로그램이 편리하게 접근해 사용할 수 있는 정의들을 포함한다. 

· 사용자 정의 타입의 정의

· 상수 또는 변수의 선언 – 초기화도 가능하다.

· 프로시저 정의

· 외부 프로시저에 대한 인터페이스 정의

· generic 프로시저의 명명과 연산자 기호의 선언

	MODULE RATIONAL_ARITHMETIC

TYPE RATIONAL

       INTEGER :: NUMERATOR

       INTEGER :: DENOMINATOR

END TYPE RATIONAL
INTERFACE OPERATOR (+)

       FUNCTION RAT_ADD(X,Y)

          TYPE (RATIONAL) :: RAT_ADD

          TYPE (RATIONAL), INTENT(IN) :: X, Y

       END FUNCTION RAT_ADD

END INTERFACE

     ...              ! and other stuff for a complete rational arithmetic facility

END MODULE RATIONAL_ARITHMETIC



모듈과 관련된 내용은 5장에서 다룰 것이다. 

11) I/O

입출력과 관련된 Fortran90의 대표적인 특징으로 NAMELIST와 Non-advancing I/O 기능 등이 있으며 5장에서 다룰 것이다.  

3. Fortran90의 객체지향성

최근의 가장 유행하는 프로그래밍 패러다임은 객체지향 프로그래밍이다. 객체지향 프로그래밍은 소프트웨어의 reusability, modularity를 좋게 하며, 어떤 타입을 가지는 객체의 처리를 위해 프로시저와 타입들에 대한 정의를 하나로 묶어(이 묶음이 하나의 객체가 된다) 오류 발생 조건을 사전에 제거한다. Fortran90은 C++와 같은 광범위한 개념의 객체지향 프로그래밍 능력을 가지고 있지는 못하지만 프로그래머가 사용하기에 충분한 객체지향성을 가지고 있다. 

다음은 Fortran90이 가지는 객체지향성을 정리한 것이며, 각각의 내용들에 간단히 소개한다. 소개되는 내용에 대한 보다 자세한 설명과 사용법, 예제 등은 3, 4, 5장에서 다루게 될 것이다.

· data abstraction : 사용자 정의 타입

· data hiding : PRIVATE과 PUIBLIC 속성 

· encapsulation : 모듈과 data hiding 특성

· inheritance와 extensibility : super-type, 연산자 overloading과 generic 프로시저
· polymorphism : generic overloading을 이용한 프로그래밍

· reusability : 모듈 

1) Data Abstraction

Fortran90은 사용자 정의 타입으로 일정 수준의 데이터 abstraction 기능을 제공하며, 포인터를 이용해 복잡한 자료구조의 정의를 가능하게 한다. Fortran90의 포인터는 다른 언어들의 그것과는 좀 달리 구현되는데, C의 포인터보다 유연성 측면에서는 부족하지만 효율적인 면에서는 오히려 더 우수하다. Fortran90의 포인터는 강형(strongly typed)
이며, 포인터의 target이 제한적이지만 이로 인해 보다 안전하고 빠른 코드 작성을 가능하게 한다. Fortran90은 열거형 타입을 지원하지는 않지만 모듈을 이용해 큰 불편 없이 구현이 가능하다. 

Fortran90에서 부족한 점은 매개변수 정의 타입(parameterized derived types)과 서브타입(subtype)에 대한 지원이 없다는 것이다. 매개변수 정의 타입은 사용자 정의 타입에서 각 타입 종류의 선택을 매개변수를 통해 할 수 있는 것을 의미한다. 예를 들어 Fortran90에서 다음에 있는 두 개의 사용자 정의 타입은 각각 독립적으로 존재하는 것이지만, 매개변수 정의 타입에서는 각 타입에 대한 skeleton만 정의하고 KIND와 같은 매개변수를 통해 선택적으로 사용할 수 있도록 한다. 

	TYPE T1

INTEGER(KIND=1) :: height

REAL(KIND=1)    :: weight

END TYPE T1

TYPE T2

INTEGER(KIND=2) :: height

REAL(KIND=2)    :: weight

END TYPE T2


서브타입은 부모 타입의 부분집합을 의미한다. 예를 들어, 기본 INTEGER 타입과 그 특성이 같지만 제한된 범위만 다른 자연수 타입을 생각해 볼 수 있다. 서브타입에 대한 구현은 어렵지만 변수의 제한범위 점검에 대한 안전성을 제공한다.  

2) Data Hiding

기본적으로 모듈 내의 모든 객체는 USE문을 가진 모든 프로그램 unit에서 접근할 수 있다. Fortran90에서는 PRIVATE 또는 PUBLIC 속성을 이용해 모듈 내의 객체나 프로시저에 대한 접근을 제한함으로써 data hiding을 지원하고 있다.

	PRIVATE                                     ! set default visibility

INTEGER :: pos, store, stack_size              ! hidden

INTEGER, PUBLIC  :: pop, push               ! not hidden



3) Encapsulation

Encapsulation은 여러 관련된 기능 혹은 객체들을 하나의 라이브러리 혹은 패키지로 정의함으로써 사용자가 복잡한 내부구조에 대한 이해 없이 관련 기능을 사용할 수 있도록 하는 것을 의미한다. Fortran90에서의 encapsulation은 data hiding, MODULE/MODULE PROCEDURE와 USE문의 사용, 그리고 모듈 내의 데이터 객체에 대한 재명명(rename), 특정 개체에 대한 선택적 접근 등에 의해 이루어 진다. 

	MODULE globals

REAL, SAVE : a, b, c

INTEGER, SAVE :: i, j, k

END MODULE globals

USE globals                      ! allows all variables in the module to be accessed
USE globals, ONLY: a, c           ! allows only variables a and c to be accessed

USE globals, r =>a, s => b        ! allows a and b to be accessed with local variables r and s



4) Inheritance and Extensibility

Fortarn90은 super-type을 지원한다. 여기서 super-type이란 다른 정의된 타입을 포함하는 사용자 정의 타입을 말하며, 이렇게 하나의 계층적 구조가 형성된다. Fortran90은 서브타입을 지원하지 않기 때문에 객체나 기능에 대한 계층상의 상속이 이루어 지지는 않는다. 상속은 객체지향 프로그래밍에서 매우 중요한 개념이지만 Fortran90에서는 이것을 지원하지 않는다. 

5) Polymorphism

Generic 프로시저에 의해 구현되는 Fortran90의 polymorphism은 다음 2장에 설명되어 있다. Fortran90의 generic 특성은 진보된 것이지만, 특정 프로시저를 항상 사용자가 작성해서 generic 인터페이스에 첨가해야 한다. 

6) Reusability

모듈은 다른 모듈에 의해 사용 가능한 함수와 자료구조의 선언 등에 대한 라이브러리를 구성하며, 이렇게 모듈에 포함된 내용은 언제든지 다른 프로그램 unit에 의해 접근될 수 있고 이를 통해 사용자는 reusability가 좋은 코드를 손쉽게 작성할 수 있다.  

4. Fortran77, C, C++와 Fortran90의 비교

탁월한 안정성과 다양한 라이브러리 지원 등으로 인해 Fortran은 계산과학 분야에서 가장 많이 사용되는 프로그램이었다. 그러나, 최근에 와서 동적 자료구조, 유닉스 워크스테이션, 대화식 가시화 능력, 병렬구조 등에 대한 중요성이 대두 되면서 C 또는 C++와 같은 새로운 언어가 계산과학 분야에서 많은 관심을 받고있다. 이에 Fortran은 Fortran90으로 진화 하면서 현대 계산과학 분야에서 필요로 하는 여러 가지 능력들을 갖추게 되었는데 다음 표는 Fortran77, C, C++와 Fortran90을 몇 가지 측면에서 비교한 것을 순위로 나타내 요약한 것이다. 

	Functionality 
	Fortran77
	C
	C++
	Fortran90

	Numerical Robustness
	2
	4
	3
	1

	Data Parallelism
	3
	3
	3
	1

	Data Abstraction
	4
	3
	2
	1

	Object Oriented Programming
	4
	3
	1
	2

	Functional Programming
	4
	3
	2
	1

	Average
	3.4
	3.2
	2.2
	1.2


표 1. 계산과학을 위한 언어들의 상대적 순위 (출처:http://www.comphys.uni-duisburg. de / Fortran90/pl/pl.html)
1) Numerical Robustness

Numeric polymorphism, kind 타입, 정밀도 선택, numeric environmental inquiry 등으로 인해 Fortran90이 1위가 됐으며, 복소수 타입의 지원으로 Fortran77이 2위, C++는 C보다 polymorphism 측면에서 우월해 그 다음 순위가 되었다. Fortran90의 numerical robustness는 2장에서 상세히 다룰 것이다. 

2) Data Parallelism

Data parallel은 Fortran90만 지원하고 있으며 이에 대해서는 3장에서 다루게 될 것이다. 

3) Data Abstraction

Fortran90은 사용하기 편리하고 효율적인 data abstraction 능력을 가지고 있다. C++는 object oriented programming의 한 부분으로서 data abstraction 능력을 가지며, 이는 계산과학 분야에서 사용하기에 Fortran90보다 복잡하다. C는 자료구조의 지원 측면에서 Fortran77보다 나은 능력을 가진다. 

4) Object Oriented Programming

Fortran90은 자동상속을 지원하지 않아 C++보다는 못하지만 polymorphic 특성의 수동상속이 가능해 Fortran77이나 C보다는 좋은 순위를 가진다. 이 부분에서도 C는 자료구조의 지원 측면에서 Fortran77보다 더 앞선다.  

5) Functional Programming

Recursion과 자료구조 지원의 부족으로 Fortran77은 최하위가 되었으며, Fortran90은 필요할 때만 값을 계산하는 lazy evaluation 기능으로 최고점을 받았고 lazy evaluation은 지원하지 않지만 polymorphism을 지원하는 C++가 C보다 좋은 점수를 받았다. 

Ⅱ Fortran 90의 Numerical Robustness
많은 계산과학 분야의 문제는 수학적 모델을 수치적으로 전산실험 하는 것이므로 수치 계산은 계산과학 분야에서 핵심적인 부분이며, 따라서 수치적 기능은 매우 중요하다. 수치적인 기능은 단정도와 배정도 실수 타입, 복소수(단정도) 데이터 타입과 다양한 수치계산 문제에서 사용할 수 있는 풍부한 함수 라이브러리 등으로 구성되며 때때로, 배정도 복소수 타입과 4배정도 실수 데이터 타입과 같은 기능이 추가되기도 한다. 대부분의 과학계산 문제는 이러한 수치적 기능만으로 충분하다.  

그러나, 경우에 따라서 위에서 언급된 수치적 기능만을 사용하는 것보다 현재 수치적 implementation 환경에 대한 더 많은 정보를 사용하거나 수치적 정밀도를 더 높여줌으로써 수렴성에 대한 보장이나 결과의 정밀도를 높이는 등의 수치계산 성능을 높일 수 있다. 여기서는 “type kind” 기구를 이용한 수치적 정밀도 선택과 수치적 implementation 환경에 대한 정보를 얻는 numeric approximation model, environmental intrinsic function 등에 대해 알아본다. 

1. Numeric Kind Parameterization

Fortran90의 KIND mechanism은 Fortran77에서 공통적으로 사용되던 “*size” 형식을 일반화, 정식화, 표준화 시킨 것이다. “*size” 문법은 서로 다른 종류의 실수 타입을 변수 size를 이용해 형식화 한 것으로 Fortran77에서 단정도 REAL은 REAL*4와 같고 DOUBLE PRECISON은 REAL*8과 동일하다. Fortran90에서는 고유의 데이터 타입마다 정수의 kind 값을 대응시키는 타입 kind 개념을 정식화 했으며 이러한 kind 값들은 implementation dependent하게 설정 되어진다. 만약 단정도 REAL에 4를 DOUBLE PRECISION에 8을 대응 시켰다면 REAL과 REAL(4), REAL(KIND=4)는 동일한 표현이며, DOUBLE PRECISION과 REAL(8), REAL(KIND=8)은 동일한 표현이다. Fortran90의 REAL 타입 선언 문법은 다음과 같다. 

	REAL [([KIND= ] kind-value)]




어떤 시스템에서는 단정도 REAL의 크기가 4바이트가 아닐 수 있으며, 혹은 단정도 REAL을 표현하는 방식도 하나 이상이 될 수도 있다. 이런 경우, Fortran90은 다른 타입의 실수 표현에는 다른 kind 값을 대응시킬 뿐이다. 모든 시스템이나 implementation에 적합한 명확한 kind 값의 집합은 없으며 따라서, kind 값은 implementation dependent하게 결정된다. 

고유함수 KIND를 사용하면 코드작성을 implementation의 kind 값 설정에 무관하도록 할 수 있다. KIND함수는 입력으로 들어온 인수에 대해 현재 implementation에서의 kind 값을 정수로 리턴해 주는 Fortran90의 고유함수 이다.  

	INTEGER, PARAMETER :: SINGLE = KIND(1.0),  &

                       DOUBLE = KIND(1D0)

REAL(KIND=SINGLE)       … single precision variables …
REAL(KIND=DOUBLE)      … double precision variables …
…



2. 정밀도 선택

Fortran90의 고유함수 SELECTED_REAL_KIND는 최소의 소수점 정밀도와 지수 범위를 알려줌으로써 사용자가 원하는 정밀도의 변수를 사용할 수 있도록 해준다. SELECTED_REAL_KIND는 사용자가 원하는 소수 정밀도와 지수범위를 나타내는 두 개의 선택적인 정수 인수를 가지는데 적어도 하나는 있어야 한다. SELECTED_REAL_KIND는 현재의 implementation에서 입력 조건에 맞는 정밀도를 지원하는 최소의 실수 데이터 타입 kind 값을 리턴 해준다. 아래의 예에서 P9은 소수 9번째 자리까지의 정밀도를 나타낼 수 있는 정수의 kind 값을 함수 SELECTED_REAL_KIND로부터 리턴 받은 것이다. 

	INTEGER, PARAMETER :: P9 = SELECTED_REAL_KIND(9)

REAL(KIND=P9)       … 9 digit (at least) precision real variables …
…



IBM 시스템에서 P9은 DOUBLE PRECISION 혹은 앞 절의 DOUBLE과 동일하다. 그러나 Cray 시스템에서는 단정도 실수 혹은 앞 절의 SINGLE과 동일하다. 

실수 상수들은 어떤 종류라도 다음과 같이 “_”에 kind 값을 붙여 나타낼 수  있다. 

	1.41_SINGLE and 1.41 are the same,

1.41_DOUBLE and 1.41D0 are the same,

1.41_P9 is the appropriate 9+ digit representation of 1.41, and  typically will be equivalent to 1.41_SINGLE(Cray) or 1.41_DOUBLE(IBM)




3. Numeric Polymorphism

Fortran90의 모든 계산관련 고유함수 들은 implementation에 의해 제공되는 모든 타입에 대해 generic하다. 예를 들자면, 고유함수 COS(X)의 결과는 인수 X가 SINGLE, DOUBLE 또는 P9이냐에 따라 각각 SINGLE, DOUBLE, P9으로 리턴된다. Fortran77에서도 고유함수는 역시 generic 하지만 사용자나 혹은 다른 벤더에 의해 제공되는 프로시저는 generic 하지 않다. Fortran90은 이러한 Fortran77의 결점을 보완하고 있다. 여기서 말하는 Polymorphism은 사용자 정의 프로시저까지 확장된 Fortran90의 generic 속성을 의미한다. 

사용자가 만들어 제공하는 프로시저 집합에 대해 generic 이름을 붙이거나 기존의 generic 이름에 새로운 프로시저를 추가하기 위해 interface block을 사용한다. 서로 다른 타입의 인수를 가지는 프로시저들을 generic interface로 묶어 사용하고 generic으로 묶인 프로시저를 프로그램에서 호출할 때는 하나의 generic 이름으로 호출하지만 입력 인수 타입에 의해 각 프로시저가 구분된다. 

· INTEGER와 REAL을 swap하는 external subroutine

	SUBROUTINE swapint (a, b)
INTEGER, INTENT(INOUT) :: a, b

INTEGER :: temp

temp = a; a = b; b = temp

END SUBROUTINE swapint
SUBROUTINE swapreal (a, b)
REAL, INTENT(INOUT) :: a, b

REAL :: temp

temp = a; a = b; b = temp

END SUBROUTINE swapreal



· generic swap routine “swap”을 만들고 프로그램에서는 generic 이름 “swap”으로 호출한다. 이때 인수 타입이 generic interface의 어떤 프로시저를 호출하는지를 결정한다. 

	INTERFACE swap                       ! generic name

SUBROUTINE swapreal (a,b)

REAL, INTENT(INOUT) :: a, b

END SUBROUTINE swapreal
SUBROUTINE swapint (a, b)

INTEGER, INTENT(INOUT) :: a, b

END SUBROUTINE swapint

END INTERFACE
!main program

INTEGER :: m,n

REAL :: x,y

...

CALL swap(m,n)

CALL swap(x,y)



다음의 예는 interface block을 이용하여 네 개의 프로시저를 묶어 새로운 generic 이름(SMOOTH)를 정의한 것이다. 인수 AA는 각 경우마다 타입이 다르지만 함수의 결과는 모두 INTEGER로 동일하다. 그렇지만, 함수의 결과는 모든 경우 혹은 일부분에 대해 다른 타입을 가질 수도 있다. 

	INTERFACE SMOOTH                           ! SMOOTH is the generic name

INTEGER FUNCTION SMOOTH_INT(AA)         ! for procedures   SMOOTH_INT

     INTEGER :: AA(:,:)                        ! SMOOTH_SINGLE

END FUNCTION SMOOTH_INT                 ! SMOOTH_DOUBLE

                                              ! SMOOTH_RATIONAL

INTEGER FUNCTION SMOOTH_SINGLE(AA)

     REAL(SINGLE) :: AA(:,:)                    ! AA is an assumed shape two-
END FUNCTION SMOOTH_SINGLE              ! dimensional array in each case.
INTEGER FUNCTION SMOOTH_DOUBLE(AA)

     REAL(DOUBLE) :: AA(:,:)

END FUNCTION SMOOTH_DOUBLE
  INTEGER FUNCTION SMOOTH_RATIONAL(AA)

     TYPE(RATIONAL) :: AA(:,:)

  END FUNCTION SMOOTH_RATIONAL

END INTERFACE



이미 존재하는 generic 이름에 새로운 프로시저를 추가할 수도 있다. 다음 예는 고유함수 COS에 인수 타입 RATIONAL에 대한 연산 프로시저를 추가한 것이다. 

	INTERFACE COS                          ! Extends the generic properties

   FUNCTION RATIONAL_COS(X)           !  of COS to return results of

     TYPE(RATIONAL) :: RATIONAL_COS    !  type RATIONAL, assuming the

     TYPE(RATIONAL) :: X                 !  argument is of type RATIONAL.

   END FUNCTION RATIONAL_COS 

END INTERFACE



4. Numeric Approximation Model

이식 가능하고 수치적으로 robust한 프로그램 개발을 위해 implementation의 numerical 특성에 동적으로 접근할 수 있도록 하는 것이 중요하다. Fortran90은 실수 근사 모델과 모델 관련 값을 알아볼 수 있는 16개의 고유함수를 제공해서 그러한 접근이 가능하도록 하고 있다. 

Fortran90에서 각 종류의 실수들은 다음과 같이 모델링 되어 진다. 

where 
x is the real value 

s is (the sign of the value) 

b is the radix (base) and is usually 2; b is constant for a given real kind 

p is the base b precision; p is constant for a given real kind 

e is the base b exponent of the value 

각 실수 타입에 대한 주요 특징은 b 와 p 그리고 e 의 범위에 의해 결정되는데 대부분 implementation에서 서로 다른 실수 종류는 p에 의해 결정된다. 
IEEE arithmetic
· b  = 2 : binary representation

· p  = 24 : single precision

· p  = 56 : double precision

· -127< e <127
· e  = -127 : 0과 NaNs(illegal or out-of-range value) 표현에 사용
IBM 370 real arithmetic : non-binary example

· b  = 16 : binary representation

· p  = 6 : single precision

· p  = 14 : double precision

· -127< e <127
5. Environmental Inquiry

각 종류의 실수를 특징짓는 미리 정의된 9개의 모델 관련 값이 있다. 사용자는 environmental inquiry 고유함수를 이용해 데이터 objects의 선언이나 기타의 계산에서 필요한 경우 언제든지 이 값들에 접근할 수 있다. 각 함수는 단 하나의 인수를 가지는데 이 인수는 상수이거나 스칼라 변수 또는 배열이 될 수 있다. 

	Characteristic values of a real kind
	Intrinsic function name

	the decimal precision
	PRECISION

	the decimal precision range
	RANGE

	the largest value
	HUGE

	the smallest value
	TINY

	a small value compared to 1; b1-p
	EPSILON

	the base b
	RADIX

	the value of p
	DIGITS

	the minimum value of e
	MINEXPONENT

	the maximum value of e
	MAXEXPONENT


표 2. Environmental intrinsic functions

	REAL :: a

PRINT *, 'PRECISION =', PRECISION(a)

PRINT *, 'HUGE =', HUGE(a)

PRINT *, 'RADIX =', RADIX(a)

PRINT *, 'DIGITS', DIGITS(a)

PRINT *, 'MINEXPONENT =', MINEXPONENT(a)

…



다음은 위의 코드를 KISTI IBM시스템에서 실행한 결과이다. 

	$a.out

PRECISION = 6

HUGE = 0.3402823466E+39

RADIX = 2

DIGITS 24

MINEXPONENT = -125



7개의 numeric manipulation 함수를 이용해 사용자는 인수로 주어진 수의 종류와 관련된 모델 값에 접근할 수 있다. 7개 중 3개는 2개의 인수를 나머지 4개는 하나의 인수를 가진다. 

	Values related to the argument value
	Intrinsic function name

	exponent value of the number, e
	EXPONENT(X)

	fractional part of the number, 

	FRACTION(X)

	returns the nearest representable number(second argument specifies direction)
	NEAREST(X,S)

	returns the inverted value of the distance between the two nearest possible numbers
	RRSPACING(X)

	multiplies X by the base to the power I (change e by value of the second argument)
	SCALE(X,I)

	returns the number that has the fractional part of X and the exponent I (set e to value of second argument)
	SET_EXPONENT(X,I)

	the distance between the two nearest possible numbers
	SPACING(X)


표 3. Numerical manipulation functions

	PRINT*, 'EXPONENT =', EXPONENT(10.2)

PRINT*, 'FRACTION =', FRACTION(10.2)

PRINT*, 'NEAREST =', NEAREST(3.0, 2.0)

PRINT*, 'SPACING =', SPACING(3.0)

…



다음은 위의 코드를 KISTI IBM시스템에서 실행한 결과이다.

	$a.out 

EXPONENT = 4

FRACTION = 0.6374999881

NEAREST = 3.000000238

  SPACING = 0.2384185791E-06


다음은 Newton 방법에서 최적의 정밀도를 얻기 위해 SPACING 함수를 이용한 예이다. 

	…
DO

DX = F(X)/DF(X)                  ! Compute the next delta-X

X = X-DX

IF (DX<2*SPACING(X)) EXIT       ! Stop if near the spacing limits of 

ENDDO                            ! that kind in this region

…



Ⅲ POWER4 시스템 성능 및 튜닝 
1. 튜닝 과정
프로그램의 튜닝단계를 중요성에 따라 정리하면 다음과 같다. 새로운 프로그램을 만들 때 이러한 가이드라인을 따라 간다면 나중에 튜닝하는데 드는 시간을 절약할 수 있다.
1. 프로그램에서 I/O가 중요한 부분을 차지 한다면 I/O 튜닝을 computational 튜닝과 분리해야 한다.

2. 컴파일러 최적화 옵션 중 최적의 옵션들을 이용한다.

3. 프로파일링을 통해 프로그램 내에서 핫스팟을 찾아 낸다. 이 과정은 프로그램을 튜닝할 때 시간낭비를 막을 수 있다.

4. 가능하다면 MASS, ESSL 혹은 성능이 최적화된 라이브러리를 사용한다.

5. 일반적인 튜닝 기준을 따른다.
6. 프로그램을 POWER4에 맞게 핸드튜닝한다.
2. 핸드 튜닝

p690 시스템 같은 캐시에 기초한 RISC 아키텍쳐 컴퓨터에의 핸드 튜닝은 다음과 같이 2부분으로 나눌 수 있다.

1. Avoid the negative

computational unit보다 느린 캐시나 메모리 서브시스템으로의 액세스를 피하거나 줄인다. 이를 위한 기본적인 방법은 다음과 같다.
· Stride minimization

· Encouragement of data prefetch streaming

· Avoidance of cache set accociativity constraints

· Data cache blocking

2. Exploit the positive

computational unit 특히 floating-point unit을 효율적인 활용율을 최대화하는데 이러한 CPU 튜닝 방법은 다음과 같다.

· Unrolling inner 루프 - 파이프라인을 채운 상태로 유지하기 위해 각 iteration에서 독립적은 계산량을 늘인다. 
· Unrolling outer 루프 – computation 대 load/store instruction의 비율을 증가시켜 루프 성능이 데이터 전송보다는 computation에 의해 제한받도록 한다.
3. POWER4 칩의 구조
3.1 Level 1,2,3 캐시

각 캐시에 대한 간단한 설명은 다음과 같다.

3.1.1 Level 1 instruction 캐시
L1 instruction 캐시(I-cache)는 64KB이며 이는 트랜잭션 프로세싱과 같은 상용 어플리케이션에서 아주 중요하다. computation 중심의 어플리케이션은 상대적으로 적은 instruction의 아주 긴 루프(Fortran의 DO-루프, C의 for-루프)로 구성되어 있기 때문에 L1 instruction 캐시가 성능에 있어서 크게 영향을 주지 않는다. L1 instruction 캐시에 대한 튜닝은 루프가 아주 많은 수의 instruction을 포함하지 않도록 하는 것이다.
3.1.2 Level 1 데이터 캐시
각 프로세서는 32KB의 독립된 L1 데이터 캐시를 가지고 있다. 이는 가장 먼저 저장된 것이 가장 먼저 교체되어 나가는(FIFO) 알고리듬에 의해 동작되며 2-way set associative이다.
3.1.3 Level 2 캐시
각각의 POWER4 칩은 1440KB의 독립된 L2 캐시(instruction + 데이터)를 가지고 있다. IBM pSeries 690 Model 681 및 pSeries 690 Turbo 시스템은 하나의 칩에 2개의 프로세서를 가지고 있으며 각 프로세서는 L2 캐시를 공유하고 있다. 또한 pSeries 690 HPC 시스템은 하나의 칩에 1개의 프로세러를 가지고 있기 때문에 독립된 L2 캐시를 갖는다. 
3.1.4 Level 3 캐시
4개의 POWER4 칩이 하나의 MCM(Multi-chip Module)을 이루며 이는 128MB의 L3 캐시를 가지고 있다. pSeries 690 Model 681 시스템은 하나 이상의 MCM으로 구성되어 있으며, 적당한 성능감소를 가지면서 다른 MCM의 L3 캐시를 액세스할 수 있는데, 이는 LPAR로 파티셔닝된 시스템에도 적용된다.

3.2 일반적인 캐시의 특징

L2 캐시에서 L1 캐시로의 높은 bandwidth는 FPU로의 데이터 공급에 있어서 충분히 남는크기이다. 성능 측면에서 L1 캐시와 L2 캐시의 중요한 차이점은 latency이다. 부동소수점 리지스터와 L1 캐시사이의 load/store는 약 4사이클의 latency를 가지고, 리지스터와 L2 캐시 사이에서는 약 14사이클의 latency를 가진다. 

촘촘한 matrix같은 것을 다루는 계산문제에서 추천되는 튜닝방법은 L2 캐시를 위한 데이터 blocking과 L1 캐시를 위한 연속적인 데이터 액세스를 하는 것이다.
3.3 L1 데이터 캐시의 구조
3.3.1 캐시라인(Cache Lines)
메모리는 연속적인 128byte 라인들로 나누어져 있고, 캐시도 이와 비슷하게 나누어져 있으며, 캐시와 메모리 사이에서 모든 데이터의 전송이 이 라인을 단위로 이루어 진다. 예를 들어 만약에 어떤 부동소수점수가 계산에 사용되기 위해 부동소수점 리지스터로 복사(load)되어야 할 경우 이 수를 포함하는 전체 캐시라인이 메모리에서 캐시로 전달 되어 진다.

POWER4 L1 데이터 캐시

3.3.2 Set associativity
L1 데이터 캐시는 위 그림과 같이 메모리에 매핑된다. 각 컬럼을 congruence class라고 부르며, 메모리상의 임의 라인은 동일 congruence class에 있는 캐시의 두 라인 중 하나에 복사(load)된다. 또한 POWER4 L1 데이터 캐시는 128개의 congruence class를 가진 2-way set assocative이며 각 캐시라인이 128byte이기 때문에 총 L1 데이터 캐시는 32768byte가 된다. 새로운 라인이 L1에 복사(load)될 때에는 congruence class의 두 라인중 오래된것과 대체된다. (FIFO replacement)
POWER4 캐시에서 가장 나쁜 경우는 16KB 혹은 16KB의 배수 크기로 데이터 element들이 펼쳐져 있는 것이다. 이 경우 모든 element들이 같은 congruence class에 놓이게 되고, effective cache size가 두라인 밖에 되지 않는다. 

3.4 L2 캐시의 특징

L2 캐시 크기는 POWER4 칩당 1440KB이고, 이는 각 칩의 두 프로세서가 공유하고 있다. L1 데이터 캐시와 마찬가지로 캐시라인 크기는 128byte이다. L2 캐시는 독립적인 L2 캐시 컨트롤러에 의해 제어되는 3개의 동일한 부분으로 나누어 진다. 16개의 연속적인 Fortran double-precision 배열 element(128bytes)는 동일한 캐시라인에 놓이며, 같은 캐시 컨트롤러에 의해 제어되지만, 17th element는 다른 캐시라인에 놓이게 되고, 또한 hashing 알고리듬에 의해 다른 캐시 컨트롤러에 의해 제어된다. 이는 array를 연속적으로 엑세스할때 저장과정의 최적화를 보여준다.
3.5 FPU(Floating Point Units)

pSeries 690 Model 681 시스템의 프로세서에서 부동소수점 비율을 가능한 최대로 하기 위해서는 메모리 서브시스템에 의한 지연을 없애고, 프로그램이 L1 캐시에 놓이도록 해야 한다. 

POWER4 프로세서의 구조

FPU의 특징을 정리하면 다음과 같다.

1. pSeries 690 Model 681 시스템은 하나의 프로세서당 2개의 FPU가 있고, 이들은 하나의 L1 캐시를 서로 공유하고 있으며 독립적으로 작동된다. 
2. 두개의 load/store unit에 의해 데이터는 L1 캐시로부터 리지스터로 복사되고(load), 리지스터로부터 L2/L1 캐시로 저장된다(store).
3. L1/L2 캐시의 데이터에 대해서 부동소수점 double-precision(REAL*8) 변수의 load/store 과정은 한 사이클당 한번의 비율로 load/store unit에 의해 이루어진다. 여기서 load 과정인 경우 계산에 있어서 FPU가 사용하려는 데이터의 위치에 따라 다른 latency를 보인다. 데이터가 L1 캐시에 있는 경우 대략 4 사이클의 latency가 생기고, L1 캐시에 없고, L2 캐시에 있는 경우 14 사이클의 latency가 생긴다. 컴파일러는 L1 latency(4 사이클)로 가정하기 때문에 성능을 최적화하기 위해서는 load되는 데이터가 L1에 위치해 있는지의 여부가 중요하다.
4. 기본적인 부동소수점 instruction은 double-precision multiply/add, multiply/subtract, negative multiply/add, negative multiply/subtract 등의 연산이 있다. multiply/add는 2개의 부동소수점 연산이며, 하나의 add, subtract, mulitply는 multiply/add와 동일한 하드웨어를 사용하면서 같은 시간이 소요된다. 그래서 어떤 프로그램이 add만 수행한다면 이는 multiply/add를 수행하는 프로그램에 비해 절반정도의 MFLOPS로 계산이 이루어진다.
5. 나눗셈은 파이프라인화 되지 않고, 아주 많은 시간이 소요된다.
4. 메모리 시스템에 대한 튜닝
4.1 Stride minimization

하나의 라인이 캐시로 load되면 동일 캐시라인에 있는 모든 다른 피연산자도 참조된다. 만약 데이터가 넓은 영역에 퍼져 있다면 load된 캐시라인에서 참조되는 피연산자가 적을 것이며, 특히 double precision인 배열 데이터의 값들이 16보다 크게 퍼져 있거나 single precision인 배열데이터의 값들이 32보다 크게 퍼져 있으면 load된 각각의 캐시라인에서 단 하나의 데이터 값만 참조될 것이다. 그래서 load된 캐시라인의 데이터 값들이 나중에 엑세스될 때 캐시에 더 이상 있지 않을 것이며 이는 곧 캐시 miss의 overhead로 이어 진다. 그래서 순차적인 데이터 엑세스를 통해 이런 경우를 피해야 할 것이다.
 fortran 배열은 메모리에서 column 방향으로 저장이 되고, C 배열은 row 방향으로 저장이 된다. 이를 고려하여 다중배열에서 데이터를 순차적으로 엑세스하도록 하는 것은 가장 기본이 되는 튜닝 기술이다. 

Fortran 예제

	do i=1,n

do j=1,n

a(j,i)=a(j,i)+b(j,i)*c(j,i) ! Left subscript same as inner 루프 var.

enddo

enddo



C 예제

	for(i=0;i<n;i++)

for(j=0;j<n;j++)

{a[i][j]+=b[i][j]*c[i][j];}  /* Right index same as inner 루프 variable */



위와 같은 간단한 경우 컴파일러가 알아서 루프 순서를 바꾸지만 코딩을 할 때 컴파일러에 의존하기 보다는 항상 바른 루프 순서로 코딩을 해야 할 것이며, 일반적으로 코딩을 하다 보면 위의 예와 같이 모든 배열을 항상 순차적으로 엑세스할 수 없는 경우가 많이 있다.

4.2 Data Prefetch

POWER4 프로세서 하드웨어는 데이터 prefetching이 구현되어있다. 즉 load instuction에서 캐시 miss가 일어나면 load instruction에 의해 참조되기 전에 prefetch 엔진이 다음 캐시 라인을 엑세스하도록 가동된다. 이는 곧 데이터가 L1 데이터 캐시에 있도록 하기 위해서 L3에서 L2로, 메모리에서 L3로 prefetch되는 것이다. 다음 그림은 prefetching 과정의 순서를 보여준다. 

이러한 prefetching은 효력을 내기 위해 소프트웨어가 무언가를 지원해 줘야 하는 것은 아니다. 그러나 하드웨어 prefetching 엔진의 능력을 최대한 발휘할 수 있도록 튜닝을 하면 해당 어플리케이션의 성능이 많이 향상된다.
POWER4 데이터 prefetching은 4~8개 스트림의 루프에 최적화 되어 있다. 다음 그림은 1~8개의 stream을 가진 각각의 루프에 대한 성능을 나타내는데, 그림에서 보듯이 루프당 스티림의 수가 1개에서 8개로 늘어날수록 최대 70%까지 성능이 향상되고, 스티림의 수가 1개에서 4개로 증가할 때 향상되는 폭이 크게 나타난다.

스트림의 수를 늘이는 방법은 이웃하는 루프와 합치거나(Fusing, XLF 컴파일러의 –qhot 옵션이 하는 역할) 루프를 중앙에서 잘라 루프 크기를 줄이는 것(Midpoint bisection)이다.

Fusing 예제

	DO I=1,N

S = S + B(I) * A(I)

ENDDO

DO I=1,N

R(I) = C(I) + D(I)

ENDDO

다음과 같이 수정:
DO I=1,N

S = S + B(I) * A(I)

R(I) = C(I) + D(I)

ENDDO



Midpoint bisection 예제

	DO I=1,N

S = S + B(I) * A(I)

ENDDO

다음과 같이 수정:

NHALF = N/2

S0=0.D0

S1=0.D0

DO I=1,NHALF

S0 = S0 + A(I)*B(I)

S1 = S1 + A(I+NHALF)*B(I+NHALF)

ENDDO

IF(2*NHALF.NE.N) S0 = S0 + A(N)*B(N)

S = S0+S1



루프내의 스트림이 8개보다 많을 경우 해당 루프의 스트림 수를 줄이는 것이 전체적인 성능을 높일 수 있는데, 이는 prefetching 큐가 8개인 것처럼 단지 8개의 스트림만 prefetching될 수 있기 때문이다. 루프의 스트림 수를 줄이기 위해 8개 혹은 그보다 적은 수의 스트림을 가지도록 루프를 2개 혹은 그 이상으로 분리하는 방법이 있다. 이렇게 루프를 분리하고자 할 때 임시 배열을 만들어야 하는 경우가 발생할 수도 있는데, 어떤 경우에라도 항상 루프내의 스트림의 수를 증가 혹은 감소시키는 튜닝이 전체적인 성능향상에 효과가 있는지를 체크하기 위해 해당 어플리케이션의 프로파일링과 루프의 계산시간을 확인해야 한다.

5. FPU 튜닝

메모리 시스템에 대한 튜닝과는 달리 FPU에 대해서는 컴파일러가 많은 부분을 알아서 튜닝해 주는 편이고, 몇몇 예외 사항을 제외한다면 핸드 튜닝할 부분이 그렇게 많지 않다.

5.1 Outer 루프 unrolling

load/store 수가 FMA 수보다 많을 경우(load/store bound) 루프에서 FMA 대 load/store의 비율을 증가시킴으로서 현저한 성능향상을 가져올 수 있는데, 이는 data re-use(한번 load한 데이터를 다시 사용)인 경우로써 그 기본적인 방법이 outer 루프 unrolling이다.

다음 예제를 보면 루프가 잘 정리되어 있고 inner 루프는 스트라이핑이 잘 되어 있다.
	do i = 1, n
do j = 1, n
  y(i) = y(i) + x(j) * a(j,i)
enddo

enddo




즉 스칼라 값인 y(i)가 하나의 리지스터 안에 놓여질 수 있고, inner 루프가 완료된 후 단 한번 저장(store)되기 때문에 inner 루프에서 필요로 하는 load/store 수가 최소화되어 있다. inner 루프에서의 iteration은 store는 없이 두번의 load(x(j), a(j,i))만 필요로 한다. 만약 루프의 순서가 바뀌어 i가 inner 루프라면 두번의 loadl(y(i), a(j,i)와 한번의 store(y(i))가 필요할 것이며 추가로 a(j,i)에 대해 스트라이핑이 제대로 되지 않을 것이다.
그러나 이 루프는 FMA 수보다 load/store instruction 수가 더 많으므로, 루프의 성능은 load/store unit이 작동되는 비율에 의해 제한을 받게 된다.
j에 대한 inner 루프에 대해서 컴파일러가 자동적으로 unrolling을 할 수 있지만 이것이 FMA 대 load/store의 비율을 변경하는 것은 아니며, 이런 경우에 해법은 i에 대한 outer 루프 unrolling이다. 위 예제에 나온 간단한 루프는 –qhot 옵션을 주면 컴파일러가 코드를 최적화할 수도 있으나, 일반적으로 핸드튜닝에 의한 outer 루프 unrolling이 보다 신뢰할 수 있다.

다음 예제는 depth 4의 루프 unrolling을 보여 준다.
	do i = 1,n,4

s0 = y(i)

s1 = y(i+1)

s2 = y(i+2)

s3 = y(i+3)

do j = 1,n

s0 = s0 + x(j)*a(j,i)

s1 = s1 + x(j)*a(j,i+1)
s2 = s2 + x(j)*a(j,i+2)
s3 = s3 + x(j)*a(j,i+3)

enddo
y(i) = s0
y(i+1) = s1
y(i+2) = s2
y(i+3) = s3
enddo




위 예제에서는 element x(j)가 inner 루프에서 3번 재사용(re-used)되었기 때문에 load/store대 FMA 비율이 감소하였으며, 4번의 iteration에 대해 8번에서 4번의 load로 감소되었다.(x(j), a(j,i), a(j,i+1), a(j,i+2), a(j,i+3)) 이와 같이 unrolling depth가 증가할수록 load/store 대 FMA 비율이 2에서 1로 점근적으로 감소하게 된다. 
실제적인 성능은 컴파일러에 최적화 옵션과 핸드 튜닝에 의한 unrolling depth에 따라 다르며 POWER4 프로세서를 장착한 1.1GHz 시스템에서 수행된 몇몇 결과를 그림으로 나타내면 다음과 같다.

� IBM XLFortran에서 한 문장의 최대 길이는 6700 문자 


� IBM XLFortran은 250자 까지 사용 가능


� END DO는 Fortran77 표준이 아니다. 


� 모든 객체의 형을 정적으로 정하고 형 일치 및 그에 대한 규칙을 완벽하게 정의하여 번역시에 형 오류를 검출할 수 있는 프로그래밍 언어. 

